

Into the Silent Practice

for tenor saxophone and live electronics
Jerod Sommerfeldt
2013

Into the Silent Practice is a guided meditation for tenor saxophone and electronics. Moving from biting, angular rhythms to long tones and trills, the saxophone is supported by real-time digital signal processing that too shifts from abrasive textures to an atmospheric sound world that strives for equanimity.

Performance directions for the saxophone:

Each system in the score represents 15'' of real-time duration. Rhythms are open and free to interpretation, provided that each system is completed in the requisite 15''. With four systems to each page, one complete page of the score thus lasts 60''. A chronometer or favorite stopwatch app is recommended to coordinate with the computer.


Performance directions for the computer:

The work is realized using live, interactive electronics controlled by the laptop performer. The following equipment is necessary for accurate performance:

- (2) Microphones with boom stands, placed above and to the side of the saxophone
- Laptop with audio interface, connected to the microphones
- (2) Active PA speakers, connected to the audio interface
- Requisite cabling

The electronics part is written in RTcmix and realized using either Pure Data or Max/MSP. The performance patches - with directions for execution - for Into the Silent Practice are freely available by contacting the composer. The RTcmix code has been included as an appendix to the score.

Stage diagram:



Into the Silent Practice

Tenor saxophone and interactive electronics




Jerod Sommerfeldt

First staff of music. It begins with a treble clef and a key signature of one sharp (F#). The melody consists of eighth and quarter notes, some beamed together. There are dynamic markings: *ff* at the start, *fff* in the middle, and a final *ff* at the end. There are also some accents and slurs.

Second staff of music. It continues the melody with various note values and rests. Dynamic markings include *ff* at the start, *pp* in the middle, and *f* towards the end. There are also slurs and accents.



Third staff of music. It features a *growl* marking over a series of notes. There is a *ff* dynamic marking at the end of the staff. The melody continues with eighth and quarter notes.




Fourth staff of music. It includes *tr* (trill) markings over some notes. Dynamic markings include *mp*, *pp*, *fff*, *ff*, *mf*, and *pp*. There are also *growl* and *slap* markings. The staff ends with a *pp* dynamic marking.







The first staff of music is written on a five-line staff with a treble clef. It contains a single melodic line starting on a whole note G4 (second line), followed by a whole note F#4 (third space), then a whole note E4 (third line), and finally a whole note D4 (second space). A long, smooth slur covers the entire phrase.

[illegible][illegible]





if possible, circular breathe until end



straight, even, contemplative

mf *mp* *f* *ppp* *p*

A musical staff in treble clef containing five notes. The first note is on the first line (F4). The second note is on the second line (G4). The third note is on the second space (A4). The fourth note is on the third line (B4), marked with a double bar line. The fifth note is on the third space (C5). A long slur covers the entire staff. Below the staff, dynamic markings are placed: *mf* under the first note, *mp* under the second, *f* under the third, *ppp* under the fourth, and *p* under the fifth. Between *mp* and *f*, and between *f* and *ppp*, there are wedge-shaped markings indicating crescendos and decrescendos. Above the staff, the text "straight, even, contemplative" is written with a line pointing to the slur.

A musical staff in treble clef containing two notes. The first note is on the second space (A4). The second note is on the second line (G4).

A musical staff in treble clef containing one note on the second space (A4). Below the staff, there is a sharp sign (#) on the second line.

A musical staff in treble clef ending with a double bar line. Below the staff, the word "niente" is written.

niente

```

1  /*
2  Into the Silent Practice, for tenor saxophone and electronics
3  by Jerod Sommerfeldt
4
5  *Score file 1*
6
7  inlet1 = pitch data
8  inlet2 = attack/amplitude
9  inlet3 = pan
10 */
11 rtinput("AUDIO")
12
13 pitch_data = makeconnection("inlet", 1, 200)
14 envel_data = makeconnection("inlet", 2, 45)
15 pan_data = makeconnection("inlet", 3, 0.5)
16
17 total_duration = 115
18
19 //----- REVERB on the saxophone for color and warmth
20 bus_config("FREEVERB", "in 0-1", "out 0-1")
21
22 start = 0
23 instart = 0
24 duration = 60 * 4 //a bit longer here, bleed into last gesture
25 amplitude = 0.75
26 envelope = maketable("line", 1000, 0,1.0, 0.9,1.0, 1.0,0)
27 roomsize = 0.75
28 predelay = pan_data
29 ringdur = 1.0
30 damp = 10
31 dry = 10
32 wet = 50
33 width = 100
34 FREEVERB(start, instart, duration, amplitude*envelope, roomsize, predelay, ringdur, damp, dry, wet, width)
35
36 //----- WAVETABLE
37 bus_config("WAVETABLE", "aux 0-1 out")
38 start = 0
39 duration = 0.25
40 amplitude = 2000 //absolute 0-32768 samples
41 pitch1 = 12000
42 pitch2 = 12050
43 pan = 0.5
44 num_points = irand(8, pow(2, 16))
45
46 if(num_points >= 32768){
47     num_points = 32768
48 }
49
50 waveform = maketable("wave", num_points - 1, "sine")
51 waveform_table_length = tablelen(waveform)
52
53 loop = 0.0125
54
55 for(start = 0; start < total_duration; start += loop){
56     WAVETABLE(start, duration, amplitude, pitch1, pan, waveform)
57     pitch1 = pitch_data * 5
58
59     WAVETABLE(start, duration, amplitude, pitch2, pan, waveform)
60     pitch2 = pitch_data * 4
61
62     if(envel_data > 10){
63         pitch = pitch_data * 7
64     }
65
66     WAVETABLE(start, duration, amplitude, pitch_data*3, pan, waveform)
67
68     loop = irand(0,1) / 10
69 }
70
71 //----- COMBIT
72 bus_config("COMBIT", "in 0-1", "aux 6-7 out")
73 start = 0

```

```

74 instart = 0
75 duration = total_duration
76 amplitude = 0.75
77 envelope = maketable("line", 1000, 0,0, 0.4,1.0, 0.9,1.0, 1.0,0)
78 frequency = makeconnection("inlet", 1, 100)
79 rvbtime = 1.0
80 inputchannel = 0
81 pan1 = makeLFO("sine", frequency/10, 0,0.6)
82 pan2 = makeLFO("sine", frequency/10+1.0, 0.4,1)
83 ringdur = 1.0
84 COMBIT(start, instart, duration, amplitude*envelope, frequency*5, rvbtime, inputchannel, pan1, ringdur)
85 COMBIT(start+1, instart, duration, amplitude*envelope, frequency*6, rvbtime, inputchannel, pan2, ringdur)
86
87 //----- DEL1
88 bus_config("DEL1", "aux 6-7 in", "out 0-1")
89 start = 0
90 instart = 0
91 outdur = total_duration + 10
92 amplitude = 0.8
93 deltime = envel_data / 5
94 ramp = 0.5
95 pan = makeLFO("sine", 8.025, 0,1)
96 DEL1(start, instart, outdur, amplitude, deltime, ramp, pan)
97 DEL1(start+1, instart, outdur, amplitude, deltime = 2.5, ramp, pan)
98
99 //----- REV
100 bus_config("REV", "aux 0-1 in", "aux 2-3 out")
101 start = 0
102 instart = 0
103 duration = total_duration + 20 //a bit longer, again
104 amplitude = 0.5
105 envelope = maketable("line", 100, 0,0, 25,0.5, 50,1.0, 99,1.0, 100,0)
106 type = 1
107 rvbtime = envel_data + 2
108 rvbpct = 0.7
109 inchan = 0
110 REV(start, instart, duration, amplitude*envelope, type, rvbtime, rvbpct, inchan)
111
112 //----- ELL
113 bus_config("ELL", "aux 2-3 in", "aux 4-5 out")
114 start = 0
115 instart = 0
116 duration = total_duration + 20 //following REV
117 amplitude = 1.0
118 ringdur = 0.1
119
120 pbcut = 5000
121 sbcut = 2000
122 p3 = 0 // empty for low or high pass filters, p3 is stopband for bp filter
123 ripple = 0.7 // in dB
124 atten = 90 // in dB, attenuation at stopband
125
126 ellset(pbcut, sbcut, p3, ripple, atten)
127 ELL(start, instart, duration, amplitude, ringdur)
128
129 //----- PAN
130 bus_config("PAN", "aux 4-5 in", "out 0-1")
131 start = 0
132 instart = 0
133 duration = total_duration + 20 //following from above
134 amplitude = 1.0
135 inputchannel = 0
136 pantype = 1 //0 for constant power, 1 for linear
137 pan = pan_data
138 PAN(start, instart, duration, amplitude, inputchannel, pantype, pan_data)

```

```

1  /*
2  Into the Silent Practice, for tenor saxophone and electronics
3  by Jerod Sommerfeldt
4
5  *Score file 2*
6
7  inlet1 = pitch data
8  inlet2 = attack/amplitude
9  inlet3 = pan
10 */
11
12 rate = 8
13 control_rate(rate)
14 srand()
15
16 rtinput("PDBUF", "transition")
17
18 total_duration = 120
19
20 pitch_data = makeconnection("inlet", 1, 200)
21 envel_data = makeconnection("inlet", 2, 45)
22 pan_data = makeconnection("inlet", 3, 0.5)
23
24
25 //----- STEREO
26 bus_config("STEREO", "in 0-1", "aux 0-1 out")
27 start = 0
28 instart = 0
29 inc = 0.33
30 noteduration = inc * 0.75
31 maxgain = 0
32 mingain = -30
33 panL = makeLFO("sine", 0.5, 0,0.6)
34 panR = makeLFO("sine", 0.3, 0.4,1)
35
36 //----- MOOGVCF
37 bus_config("MOOGVCF", "aux 0-1 in", "aux 2-3 out")
38
39 for(start = 0; start < total_duration; start += inc){
40     instart = irand(0,10)
41     gain = irand(mingain, maxgain)
42     amplitude = ampdb(gain)
43     envelope = maketable("line", 1000, 0,0, 0.01,1, 0.8,1, 1.0,0)
44     noteduration += 0.001
45     STEREO(start, instart, noteduration, amplitude*envelope, panL)
46     STEREO(start+3, instart, noteduration, amplitude*envelope, panR)
47
48     vcfamplitude = 2.5
49     lowcf = irand(50.0,100.0)
50     highcf = irand(1000,14000)
51     lowres = 0.1
52     highres = 0.25
53     cf = pitch_data
54     res = pan_data - 0.4
55     MOOGVCF(start, 0, noteduration, vcfamplitude*envelope, inchan=0, panL, bypass=0, cf, res)
56     MOOGVCF(start+(start/20), 0, noteduration, vcfamplitude*envelope, inchan=0, panR, bypass=0, cf, res)
57 }
58
59 //----- MIX
60 bus_config("MIX", "aux 2-3 in", "out 0-1")
61 start = 0
62 instart = 0
63 duration = total_duration
64 amplitude = 1.0
65 envelope = maketable("line", total_duration, 0,0, 30,1.0, 180,0)
66 left = 0
67 right = 1
68 MIX(start, instart, duration, amplitude*envelope, left,right)

```

```

1  /*
2  Into the Silent Practice, for tenor saxophone and electronics
3  by Jerod Sommerfeldt
4
5  *Score file 3*
6  */
7
8  rtinput("AUDIO")
9  bus_config("FREEVERB", "in 0-1", "out 0-1")
10
11  start = 0
12  instart = 0
13  duration = 480 //last from beginning of page 5 until the end
14  amplitude = 0.4
15  envelope = maketable("line", 100, 0,0, 10,1.0, 99,1.0, 100,0)
16  roomsize = 1.0
17  predelay = 0.4
18  ringdur = 15.0
19  damp = 40
20  dry = 10
21  wet = 90
22  width = 100
23  FREEVERB(start, instart, duration, amplitude*envelope, roomsize, predelay, ringdur, damp, dry, wet, width)
24
25  /*
26  The last stage of the work, an ambient track that supports the saxophone
27  */
28
29  bus_config("WAVETABLE", "aux 2-3 out")
30  bus_config("FREEVERB", "aux 2-3 in", "aux 4-5 out")
31  bus_config("NOISE", "in 0-1", "aux 4-5 out")
32  bus_config("MOOGVCF", "aux 2-3 in", "aux 4-5 out")
33  bus_config("MMODALBAR", "aux 4-5 out")
34  bus_config("MIX", "aux 4-5 in", "out 0-1")
35
36  fibonacci_array = {1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233}
37  fibonacci_length = len(fibonacci_array)
38
39  total_duration = 360
40
41  //-----WAVETABLE
42  start = 0
43  duration = 5 //slight overlap
44  amplitude = 2000
45  envelope = maketable("line", 1000, 0,0, 0.9,1, 1.0,0)
46  pan = 0.5
47  pitch_array = {"C4", "D4", "E4", "F4", "G4", "A4", "B4", "C5"}
48  pitch_length = len(pitch_array)
49
50  for(start = 0; start < total_duration; start += 1){
51      pitch = pitch_array[trand(0, pitch_length)]
52      pan = random()
53      WAVETABLE(start, duration, amplitude*envelope, cpslet(pitch), pan) //translate to Hz
54      WAVETABLE(start+1, duration, amplitude*envelope, (cpslet(pitch) - irand(0.5,5)), pan)
55  }
56
57  //-----Long reverb
58  start = 0
59  instart = 0
60  duration = total_duration
61  amplitude = 0.9
62  envelope = maketable("line", 1000, 0,0.1, 0.6,1.0, 0.9,1.0, 1.0,0)
63  room = 0.9
64  predelay = 0.05
65  ringdur = 10
66  damp = 30
67  dry = 40
68  wet = 90
69  width = 100
70  FREEVERB(start, instart, duration, amplitude*envelope, room, predelay, ringdur, damp, dry, wet, width)
71
72  //-----NOISE
73  start = 0
74  duration = total_duration
75  amplitude = 10000
76  envelope = maketable("line", 1000, 0,0, 0.7,1.0, 1,0)
77  pan = makeLFO("sine", 10.5, 0,1)
78  NOISE(start, duration, amplitude*envelope, pan)
79
80  //-----MOOGVCF (for NOISE)
81  start = 0
82  instart = 0
83  duration = total_duration
84  amplitude = 0.3
85  envelope = maketable("random", 25, "gaussian", 0,1)

```

```

86 loopfan = 0
87 pan = makeLFO("sine", 0.25, 0,1)
88 bypass = 0
89 lowcf = 50
90 highcf = 1000
91 lowres = 0.7
92 highres = 1.0
93 cf = maketable("line", "nonorm", 2000, 0,lowcf, 10*.2,lowcf, 10*.5,highcf, 10,lowcf)
94 res = maketable("line", "nonorm", 2000, 0,lowres, 1,highres, 2,lowres)
95 MOOGVCF(start, instart, duration, amplitude*envelope, loopfan, pan, bypass, cf, res)
96
97 //-----Long reverb
98 start = 0
99 instart = 0
100 duration = total_duration
101 amplitude = 0.9
102 envelope = maketable("line", 1000, 0,0.1, 0.7,1.0, 0.9,1.0, 1.0,0)
103 room = 0.9
104 predelay = 0.05
105 ringdur = 10
106 damp = 30
107 dry = 40
108 wet = 90
109 width = 100
110 FREEVERB(start, instart, duration, amplitude*envelope, room, predelay, ringdur, damp, dry, wet, width)
111
112 //-----MMODALBAR
113 start = 0
114 duration = 0.02 //seconds
115 amplitude = 30000 //absolute btwn 0-32768
116 pitch = 7.00
117 hardness = 1.0
118 position = 1.0
119 instrument = 4 //Agogo is 2
120 pan = 0.5
121
122 stagger = 0.01
123 increment = 0.75
124
125 for(start = 0; start < total_duration; start += increment){
126     duration = (fibonacci_array[trand(0, fibonacci_length)] / 10)
127     amplitude = (fibonacci_array[trand(0, fibonacci_length)] * 40)
128     pitch = pitch_array[trand(0, pitch_length)]
129     hardness = irand(0.2, 1.0)
130     MMODALBAR(start, duration, amplitude, cpslet(pitch), hardness, position, instrument = 4, 0)
131     MMODALBAR(start+stagger, duration, amplitude, cpslet(pitch), hardness, position, instrument, 1)
132     stagger += 0.5
133     pan = trunc(random())
134     duration = (fibonacci_array[trand(0, fibonacci_length)] / 10)
135     amplitude = (fibonacci_array[trand(0, fibonacci_length)] * 30)
136     MMODALBAR(start+stagger, duration, amplitude, cpslet(pitch), hardness, position, instrument, pan)
137     instrument = pickrand(1, 2, 3, 4, 5, 6, 7, 8)
138     duration = (fibonacci_array[trand(0, fibonacci_length)] / 10)
139     amplitude = (fibonacci_array[trand(0, fibonacci_length)] * 30)
140     MMODALBAR(start, duration, amplitude, cpslet(pitch), hardness, position, instrument, pan)
141     duration += 0.0002
142     increment = (fibonacci_array[trand(0, fibonacci_length)] / 10)
143     duration = (fibonacci_array[trand(0, fibonacci_length)] / 10)
144     amplitude = (fibonacci_array[trand(0, fibonacci_length)] * 50)
145     MMODALBAR(start, duration, amplitude, cpslet(pitch), hardness, position, instrument, pan)
146 }
147
148 //-----MIX
149 start = 0
150 instart = 0
151 duration = total_duration
152 amplitude = 1.0
153 envelope = maketable("line", 1000, 0,0, 0.1,1.0, 1.0,0)
154 left = 0
155 right = 1
156 MIX(start, instart, duration, amplitude*envelope, left,right)

```